DATA GENERAL
CORPORATION
Southboro,
Massachusetts 01772
(617) 485-9100

PROGRAM

Selfloading Bootstrap and Binary Loader

TAPES

Special Format: 081-000001

ABSTRACT

The selfload tape is used in conjunction with the program load feature of the Supernova to place an absolute binary loader in the highest locations of alterable storage.

083-000001-01

1. REQUIREMENTS

1.1 Memory

Supernova with 2K or larger alterable memory.

1.2 Equipment

ASR Teletype or papertape reader.

1.3 External Subroutines

None.

1.4 Other

None.

2. OPERATING PROCEDURE

2.1 Calling Sequence

The Selfloading Bootstrap and Binary Loader is used in conjunction with the PROGRAM LOAD feature of the Supernova to place an image of the Binary Loader in the highest locations of alterable memory.

The Selfloading tape is placed in the chosen input device and the device code for that device is entered in the rightmost six console switches (bits 10-15). * 12 for PTR When PROGRAM LOAD is pressed the tape will be read in and the Supernova will halt at location 00120 in alterable storage. The Binary Loader is now in memory. Execution of the Binary Loader may now be initiated by pressing CONTINUE, or putting XX777 in the switches and pressing START.

2.2 Input Format

The Bootstrap portion of the tape is formatted to meet the requirements of the PROGRAM LOAD hardware. While the hardware inputs only 41_8 words, the attached Binary Loader which is subsequently input by the Bootstrap is formatted in the same manner as the Bootstrap itself.

Blank frames are ignored until the first nonblank frame is reached. That and subsequent frames are accepted and placed in consecutive locations of storage, beginning with location 0. The first full frame of each pair is stored in the left half of a word, the second in the right, until location 40_8 is loaded. The last instruction loaded is executed. The hardware PROGRAM is now complete.

The Bootstrap then sizes memory, interprets the device code, and reads in the Binary Loader which follows the Bootstrap on the tape and is formatted in the same manner.

2.3 Output Format

The Selfloading Bootstrap and Binary Loader produce no output.

2.4 Error Returns

There are no indications of error other than disfunction: the loaded tape should halt after the last punched frame has been read with the address lights containing 00121.

Errors occurring during the use of the Binary Loader segment of the tape are covered by document number 093-000003.

2.5 State of Active Registers upon Exit

PC: last location in read/write memory (XX777).

2.6 Cautions to User

None.

3. DISCUSSION

3.1 Algorithms

The device code is appended to the input instructions by reading the console switches, masking all but the rightmost six bits, and using the result as a count in a loop which increments the input instructions which are loaded with a device code of zero.

Determination of the highest location in core is accomplished by writing and reading locations at 1K increments until the information read back is the same as that written. Loading is begun at the highest location minus the length of the loader. Load completion is detected by exhaustion of a count, which leads to a halt at 00120.

3.2 Limitations and Accuracy

None.

3.3 Size and Timing

The Selfloading portion of the loader is 41_8 locations long. Execution is faster than the input rate of all tape readers. If any delay is perceived, the loader is not being properly executed.

3.4 References

See 093-000003 for a discussion of the Binary Loader.

3.5 Flow Diagrams

None.

4. EXAMPLES AND APPLICATIONS

Not pertinent.

5. PROGRAM LISTING

Program listing follows. For a listing of the Binary Loader see the appropriate document.

```
                        ;
                        ; BOOTSTRAP PROGRAM
                        ; LOADS INTO LOCATIONS 0-37
                        ;
BUT NOTE
LOADS INTO
0 → 37
        000100                  .LOC 100                ;MUST BE .GE. 16 FOR MKSAVE

        060100                  OPC1 = NIOS 0
        063600                  OPC2 = SKPDN 0
        060500                  OPC3 = DIAS 0,0

00100 062677 BEG:       IORST                   ;RESET ALL I/O
00101 060477            READS 0                 ;READ SWITCHES INTO AC0
00102 024026            LDA 1,C77               ;DEVICE MASK
00103 107400            AND 0,1                 ;DEVICE CODE
00104 124000            COM 1,1                 ;-DEVICE CODE-1
        000005 LOOP=.-BEG
00105 010014            ISZ OP1                 ;ADJUST ALL
00106 010030            ISZ OP2                 ;I/O INSTRUCTIONS
00107 010032            ISZ OP3                 ;FOR PROPER CODE
00110 125404            INC 1,1,SZR             ;DONE ?
00111 000005            JMP LOOP                ;GO BACK
00112 030016            LDA 2,C377              ;JMP 377 INTO LOCN. 377
00113 050377            STA 2,377               ;...
        000014 OP1=.-BEG
00114 ~~060077~~ 060110         OPC1-1          ;NIOS DEVICE
00115 101102            MOVL 0,0,SZC            ;TEST BIT 0, CLEAR CARRY
        000016 C377=.-BEG
00116 000377            JMP 377                 ;CHANNEL DEVICE- GO WAIT FOR THR
        000017 LOOP2=.-BEG
00117 004030            JSR GET+1               ;GET A FRAME
00120 101065            MOVC 0,0,SNR            ;IGNORE ZEROS
00121 000017            JMP LOOP2               ;GO BACK
        000022 LOOP4=.-BEG
00122 004027            JSR GET                 ;GET A FULL WORD
00123 046026            STA@ 1,C77              ;STORE THRU AUTO INC.
00124 010100            ISZ 100                 ;BUMP COUNT
00125 000022            JMP LOOP4               ;READ ANOTHER
        000026 C77=.-BEG
00126 ~~000377~~ 000123 JMP 77                  ;JMP TO LAST LOCATION LOADED
        000027 GET=.-BEG
00127 126420            SUBE 1,1                ;CLEAR AC, SET CARRY
        000030 OP2=.-BEG
        000030 LOOP3=.-BEG
00130 ~~063577~~ 063610         OPC2-1          ;SKPDN DEVICE
00131 000030            JMP LOOP3               ;WAIT FOR UNDONE
        000032 OP3=.-BEG
00132 060477 060510             OPC3-1          ;DIAS 0,DEVICE
00133 107363            ADDCS 0,1,SNC           ;ACCUMULATE TWO FRAMES
00134 000030            JMP LOOP3               ;GO BACK FOR SECOND
00135 125300            MOVS 1,1                ;SWAP
00136 001400            JMP 0,3
00137 000000            0                       ;PAD TO LOC 37
00140 000000            JMP 0                   ;SUPERNOVA STARTUP

                        .END
```

BIN. LOADER STARTS AT 17636 OCTAL